

## excel yourself

Autoren: Ralf Sowa, Christian Hapke

Beachten Sie unsere [Hinweise](#) und [Nutzungsbedingungen](#). Vorgestellte Musterlösungen basieren auf MS-Excel® 2003; sie gelten ausschließlich für aufgezeigte Beispieldaten. Bitte melden Sie uns etwaige Fehler in unseren Informationen – Ihr Feedback ist willkommen: [urs.toolbox@urs-beratung.de](mailto:urs.toolbox@urs-beratung.de)

Eine Übersicht zu unseren Excel-Informationen finden Sie hier: [www.urs-beratung.de/toolbox.htm](http://www.urs-beratung.de/toolbox.htm)

### Excel - Funktionen 7

#### INDIREKT - ADRESSE

## INDIREKT

Die Funktion INDIREKT steht in zwei *Varianten* zur Verfügung:

- INDIREKT(Bezug;WAHR)
- INDIREKT(Bezug;FALSCH)

### INDIREKT(Bezug;WAHR)

Die Funktion INDIREKT ermöglicht den Zugriff auf Zellen, deren Bereichsangabe als Text vorliegt.

Ein einfaches Beispiel: In Spalte A sind mehrere Personen aufgeführt, in Spalte B sind diesen Werte zugeordnet.

	A	B	C	D	E	F
1	Name	Punkte				
2	Paul	5			Willi	=INDIREKT("A4")
3	Hella	2				
4	Willi	3				
5	Christian	10				
6	Ralf	8				
7	Fritz	7				
8	Klara	7				

In Zelle E2 steht diese Formel

**=INDIREKT("A4")**

(beachten Sie die Anführungszeichen). Die Funktion stellt einen Bezug zur Zelle A4 her. Ergebnis ist also der Inhalt von Zelle A4, hier der Text „Willi“.

Zugegeben – das ist noch nicht besonders spannend. Aber schauen wir weiter...

**INDIREKT mit „starrem“ Bezug**

Im folgenden Bild haben wir zusätzlich in Zelle E1 einen Bezug zur Zelle A4 hergestellt, die Formel lautet

**=A4**

	A	B	C	D	E	F
1	Name	Punkte			Willi	=A4
2	Paul	5			Willi	=INDIREKT("A4")
3	Hella	2				
4	Willi	3				
5	Christian	10				
6	Ralf	8				
7	Fritz	7				
8	Klara	7				

Sortieren wir die Daten aufsteigend nach Punkten in Spalte B (ohne die Formel in E2 zu verändern), erhalten wir für beide Formeln als Ergebnis „Paul“.

	A	B	C	D	E	F
1	Name	Punkte			Paul	=A4
2	Hella	2			Paul	=INDIREKT("A4")
3	Willi	3				
4	Paul	5				
5	Fritz	7				
6	Klara	7				
7	Ralf	8				
8	Christian	10				

Fügen wir eine neue Zeile 3 ein, verändert sich die Formel in E1 in einen Bezug auf Zelle A5, während die INDIREKT-Funktion weiterhin auf Zelle A4 zugreift.

	A	B	C	D	E	F
1	Name	Punkte			Paul	=A5
2	Hella	2			Willi	=INDIREKT("A4")
3						
4	Willi	3				
5	Paul	5				
6	Fritz	7				
7	Klara	7				
8	Ralf	8				
9	Christian	10				

INDIREKT ist also nicht nur in der Lage, einen Bezug auf als Text benannte Zellen herzustellen, sondern auch diesen Bezug fest zu definieren.

**INDIREKT mit dynamischem Bezug**

	A	B	C	D	E	F
1	Name	Punkte				
2	Paul	5			Willi	=INDIREKT("A4")
3	Hella	2		a4	Willi	=INDIREKT(D3)
4	Willi	3				
5	Christian	10				
6	Ralf	8				
7	Fritz	7				
8	Klara	7				

In Zelle E3 steht diese Formel

**=INDIREKT(D3)**

Nun ist der Bezug kein Text, sondern ein gewöhnlicher Verweis auf eine Zelle (hier D3). Allerdings enthält D3 einen Text, und zwar „a4“. Und diesen Text „a4“ akzeptiert die Funktion INDIREKT für einen Verweis auf eine Zelle, und zwar auf die Zelle A4. Im Ergebnis erhalten wir *den gleichen* „Willi“ wie zuvor.

**INDIREKT mit VERKETTEN zur Bezugsdefinition**

	A	B	C	D	E	F
1	Name	Punkte				
2	Paul	5			Willi	=INDIREKT("A4")
3	Hella	2		a4	Willi	=INDIREKT(D3)
4	Willi	3		a		
5	Christian	10		4	Willi	=INDIREKT(D4&D5)
6	Ralf	8				
7	Fritz	7				
8	Klara	7				

In E5 haben wir nun diese Formel

**=INDIREKT(D4&D5)**

Wir *verketteten* hier die Zellinhalte aus D4 („a“) und D5 („4“) zu der Zeichenkette (String) „a4“. Und wieder erhalten wir mittels der Funktion INDIREKT einen Verweis auf Zelle A4.

**INDIREKT mit Bereichsangaben (1)**

Auch Bereichsangaben (mehr als eine Zelle) sind möglich, doch hier unterscheidet sich die Betrachtungsweise ein wenig:

Da wir in den bisherigen Beispielen nur auf eine Zelle verwiesen haben, konnte als Ergebnis der Inhalt dieser Zelle (hier "A4") wiedergegeben werden – so entstand der Willi.

Würden wir nun einfach nur in Zelle **D6**

**=INDIREKT("A2:A8")**

eingeben, würde das Ergebnis schlichtweg "Ralf" ergeben. Würden wir die gleiche Formel in Zelle **D7** eingeben, wäre das Ergebnis "Fritz". Wie Sie sehen, wird bei einem mehrzelligen Bezug immer die Zelle ausgewählt, die der aktuellen Zeile bzw. Spalte der Formelzelle entspricht. Somit entscheidet sich die INDIREKT, wir erinnern uns daran, dass Sie nur eine Zelle wiedergeben kann, für die Zelle mit dem "aktuellen" Bezug.

Siehe hierzu auch folgendes Beispiel:

	A	B	C	D	E	F	G	H	I
1	Name	Punkte						#WERT!	=INDIREKT("A2:A8")
2	Paul	5			Willi	=INDIREKT("A4")		Paul	
3	Hella	2		a4	Willi	=INDIREKT(D3)		Hella	
4	Willi	3		a				Willi	
5	Christian	10		4	Willi	=INDIREKT(D4&D5)		Christian	
6	Ralf	8						Ralf	
7	Fritz	7						Fritz	
8	Klara	7						Klara	
9								#WERT!	

Die Formel steht in den Zellen H1 bis H9 und liefert die Ergebnisse je Zeilenposition. Für H1 und H9 wird der Fehler #WERT! gemeldet, weil diese Zellenposition außerhalb des benannten Bereichs (A2:A8) liegt.

### INDIREKT mit Bereichsangaben (2)

Um innerhalb der INDIREKT tatsächlich einen ganzen Bereich zu betrachten, bedarf es einer weiteren Funktion, z. B.:

	A	B	C	D	E	F
1	Name	Punkte				
2	Paul	5			10	=MAX(INDIREKT("B2:B8"))
3	Hella	2				
4	Willi	3		b2		
5	Christian	10		b8	10	=MAX(INDIREKT(D4&":"&D5))
6	Ralf	8				
7	Fritz	7				
8	Klara	7				

Schauen wir uns die Formel aus Zelle E2 an:

**=MAX(INDIREKT("B2:B8"))**

Ausgangsbasis ist der Text "B2:B8". Ihn definieren wir mittels INDIREKT in die Bereichsangabe B2:B8 um. Anschließend ermitteln wir mit der Funktion MAX() den größten Wert aus dem Bereich - Ergebnis 10. Wie wir sehen, wird hier nun die INDIREKT anders ausgewertet. Hintergrund ist folgender: Die Funktion MAX erwartet einen Bezug – dieser wird durch die INDIREKT geliefert, jedoch nicht vorher ausgewertet, somit ergibt INDIREKT("B2:B8") auch tatsächlich den Bereich "B2:B8".

In E5 eine Variante, in der die Bereichsangabe verkettet wird.


**=MAX(INDIREKT(D4&":"&D5))**

Das kennen Sie bereits. Beachten Sie hier aber den zu ergänzenden Doppelpunkt. Denn die Funktion erwartet einen Text in der Form "B2:B8" – der Doppelpunkt muss selbstverständlich vorhanden sein.

### INDIREKT mit Bereichsangaben (3) – Namen

Wird der Bereich als NAME definiert ...

- Einfügen
- Namen
- Definieren



... kann der NAME innerhalb INDIREKT verwendet werden:

	A	B	C	D	E	F
1	Name	Punkte				
2	Paul	5				
3	Hella	2				
4	Willi	3		Maximum		
5	Christian	10		auswertung	10	=MAX(INDIREKT(D4&D5))
6	Ralf	8			10	=MAX(INDIREKT("Maximumauswertung"))
7	Fritz	7			10	=MAX(Maximumauswertung)
8	Klara	7				

## WAHR kann entfallen

In *vollständiger Schreibweise* hätten oben dargestellte Formeln stets noch ein Semikolon und ein WAHR mitgeführt – beispielsweise

```
=INDIREKT(D4&D5;WAHR)
=MAX(INDIREKT(D4&"&D5;WAHR))
```

Und anstelle des Wertes WAHR hätte hier auch eine „1“ eingetragen werden können:

```
=INDIREKT(D4&D5;1)
=MAX(INDIREKT(D4&"&D5;1))
```

Es ist schlicht nicht erforderlich: Fehlt die Angabe, gilt der Zugriff in dieser Definition per Standard (default). Und weil es auch wirklich nicht üblich ist, dieses WAHR unnötig mitzutragen, haben wir es oben gleich weggelassen. Doch ist der Gegensatz – das FALSCH – ab und zu hilfreich bzw. notwendig:

## Zeilen-Spalten-Bezug

Zeile 1 ist Zeile 1, Zeile 15 ist Zeile 15 ... aber die Spalte A können wir auch als erste Spalte beschreiben, also Spalte 1. Und Spalte C ist Spalte 3. Wir können die Spalten mit Nummern beschreiben (wie die Zeilen) und wissen, was gemeint ist. Das ist das grundlegende Prinzip der Zeilen-Spalten-Bezugsart.

Die Zelle B8 befindet sich in Zeile 8 und Spalte 2 – also ist diese Zelle B8 auch Zeile 8 Spalte 2 (Z8S2). Und die Zelle E27 können wir ebenso mit Zeile 27 Spalte 5 definieren (Z27S5).

**INDIREKT(Bezug;FALSCH)**

Mit dem FALSCH am Ende wird die Funktion INDIREKT auf diesen Zeilen-Spalten-Bezug *umgeschaltet*.

	A	B	C	D	E	F
1	Name	Punkte				
2	Paul	5			10	=MAX(INDIREKT("Z2S2:Z8S2";FALSCH))
3	Hella	2				
4	Willi	3		Z2S2		
5	Christian	10		Z8S2	10	=MAX(INDIREKT(D4&":"&D5;FALSCH))
6	Ralf	8				
7	Fritz	7				
8	Klara	7				

Schauen wir uns die Formel aus Zelle E2 an:

**=MAX(INDIREKT("Z2S2:Z8S2";FALSCH))**

Z2S2 steht für Zeile 2 Spalte 2, und das ist in sonst *üblicher* Schreibweise die Zelle B2. Z8S2 steht also für Zeile 8 Spalte 2, und das ist wiederum die Zelle B8. Der Bezug auf "Z2S2:Z8S2" führt hier also ebenso zu B2:B8.

Für die Variante in E5 ist das nicht anders.

Der Zeilen-Spalten-Bezug ist u. a. in Kombination mit der Funktion VERGLEICH() interessant: die Funktion VERGLEICH() liefert die Positionsnummer eines Suchkriteriums aus einem definierten Bereich, und diese Positionsnummer kann für den hier verwendeten Zeilen-Spalten-Bezug die benötigte Zeilen- und / oder Spaltennummer liefern.

Nicht mehr als eine kleine Spielerei: Was liefert Ihnen wohl diese Funktion?

**=INDIREKT("Z"&VERGLEICH(MAX(B:B);B:B;0)&"S"&VERGLEICH("Name";1:1;0);FALSCH)**

Innerhalb der Funktionen VERGLEICH haben wir hier den Vergleichstyp „0“ verwendet. Dort hätte ebenso ein FALSCH stehen können oder auch nichts nach dem Semikolon.

=VERGLEICH(MAX(B:B);B:B;FALSCH)  
 =VERGLEICH(MAX(B:B);B:B;)

Auch für die INDIREKT-Funktion kann hier das FALSCH durch eine Null oder „durch Nichts“ ersetzt werden.

=INDIREKT("Z"&VERGLEICH(MAX(B:B);B:B;0)&"S"&VERGLEICH("Name";1:1;0);0)  
 =INDIREKT("Z"&VERGLEICH(MAX(B:B);B:B;0)&"S"&VERGLEICH("Name";1:1;0);)

**Hinweis zum Funktionsassistenten**

Wenn Sie nach dem Semikolon kein FALSCH bzw. keine Null setzen, können Sie den Funktionsassistenten (→ Einfügen → Funktion oder Klick auf das fx links der Bearbeitungsleiste) nicht verwenden, weil dieser ein bloßes Semikolon als "keinen" Wert interpretiert. Nach Beendigung des Assistenten ist das Semikolon schlicht nicht mehr vorhanden mit der Folge, dass der jeweils für die Funktion geltende Default-Wert gilt. Und dieser Default-Wert ist für die INDIREKT- wie die VERGLEICH-Funktion jeweils der Wert WAHR. Excel lieferte folglich für INDIREKT einen Fehler (weil der Zeilen-Spalten-Bezug in Kombination mit WAHR nicht verstanden wird) und für VERGLEICH ggf. einen anderen als den gesuchten Wert oder einen Fehlerwert, weil eine aufsteigend sortierte Liste erforderlich wäre.

## ADRESSE

Die Funktion ADRESSE liefert einen Textstring auf eine Zelle. Sie verwendet fünf Argumente:

- die Zeilenangabe (eine Zahl)
- die Spaltenangabe (eine Zahl)
- die Angabe des Bezugstyps
  - Typ 1 liefert einen absoluten Bezug (**\$A\$4**)
  - Typ 2 liefert einen gemischten Bezug (**A\$4**) – relative Spalte, absolute Zeile
  - Typ 3 liefert einen gemischten Bezug (**\$A4**) – absolute Spalte, relative Zeile
  - Typ 4 liefert einen relativen Bezug (**A4**)
- den Wahrheitswert WAHR bzw. 1 oder FALSCH bzw. 0
  - für WAHR wird der Bezug in üblicher Schreibweise ("A4") geliefert
  - für FALSCH wird ein Zeilen-Spalten-Bezug ("Z4S1") geliefert
- die Angabe der Registerblattbezeichnung (fehlt sie, gilt der Bezug für das aktive Tabellenblatt – also das Tabellenblatt, auf dem sich die Formel befindet)

	A	B	C	D	E
1	Name	Punkte			
2	Paul	5		\$A\$4	=ADRESSE(4;1;1;1)
3	Hella	2		Tabelle4!\$A\$4	=ADRESSE(4;1;1;1;"Tabelle4")
4	Willi	3			
5	Christian	10		Tabelle4!\$A\$4	=ADRESSE(4;1;1;1;"Tabelle4")
6	Ralf	8		Tabelle4!A\$4	=ADRESSE(4;1;2;1;"Tabelle4")
7	Fritz	7		Tabelle4!\$A4	=ADRESSE(4;1;3;1;"Tabelle4")
8	Klara	7		Tabelle4!A4	=ADRESSE(4;1;4;1;"Tabelle4")
9					
10				Tabelle4!\$A\$4	=ADRESSE(4;1;1;1;"Tabelle4")
11				Tabelle4!Z4S1	=ADRESSE(4;1;1;0;"Tabelle4")

ADRESSE liefert einen Textstring zu einer Zelle. Und dieser Text kann mit INDIREKT ausgelesen werden – beispielsweise liefert uns diese Formel

**=INDIREKT(ADRESSE(4;1;1;1;"Tabelle4"))**

den mittlerweile *gut bekannten* „Willi“ (unter der Voraussetzung, dass das Tabellenblatt, aus dem vorstehender Auszug stammt, auch mit „Tabelle4“ bezeichnet ist!).